# Probit: Stata log likelihood iteration 0

When working with probit models in stata the first line of the output is (for a sample of 583 with 3 variables):

Iteration 0: log likelihood = -400.01203

If I understand this correctly the iteration 0 is the log likelihood when the parameter for my 3 variables = 0.

The log likelihood function I'm working from is:

$$\ln L(\boldsymbol{\alpha}, \beta|\boldsymbol{y,z,t}) = \sum\limits_{j=1}^T I_j ln[\Phi(\frac{\boldsymbol{\alpha}\boldsymbol{z_j}}{\sigma} - \frac{\beta t_j}\sigma)] + (1 - i_j) ln[1 - \Phi(\frac{\boldsymbol{\alpha}\boldsymbol{z_j}}{\sigma} - \frac{\beta t_j}\sigma)]$$

where $T =$ total # of observations and $\boldsymbol{\alpha}$ and $\beta$ are my parameters

I assumed that by setting my parameters from $\boldsymbol{\alpha}$ and $\beta$ to 0 would be the same as setting my $\Phi$ to 0 and hence the value per observation to 0.5. this would then be multiplied by my number of observations, $\ T$:

ln L["# of observations" * ln(0.5)]

I.e. 583 * ln(0.5) = -404.1048

However, as we see, I was wrong. I assume this has to do with a constant being estimated for each parameter that changes the probability slightly (roughly to 50.35%). Where is this constant coming from? what kind of estimation is being done with my parameters?

By running the example below, you can see that the constant is based on the mean of the dependent variable transformed to the probit metric.

// get some example data
sysuse auto, clear

// estimate a probit model, use the trace
// option to see the estimates at each iteration
probit foreign price mpg, trace

// compute the mean on the estimation sample
sum foreign if e(sample)

// transform that mean to the probit metric
di invnormal(r(mean))

// compute the log likelihood
gen double ll = cond(foreign,         ///
ln(r(mean)),     /// ll for foreing cars
ln((1-r(mean)))) /// ll for domestic cars
if e(sample)

// compute the sum
sum ll, meanonly
di r(sum)

• thanks a lot, that helped to understand part of the puzzle. however, what I don't fully understand is how to go from the iteration 0 constant of -.5321897 to the log-likelihood value for iteration 0 of -45.03321. I thought that would be # of observations * ln(cdfnormal(constant))? Jan 24 '14 at 16:29
• I edited the answer Jan 24 '14 at 16:37